Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Lend/Lease/Savings



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [.] [.] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⋇. When you see ⋇ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [.] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [.] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [.] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [.] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAM	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
PLE"		
02 "THIS IS	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
A "		
03 "⊢SAMPLE	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
"		
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↑*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND	[STO] [■] [.] L	*10 STO IND L*
L		
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

*Requires one memory module.

CONSTANT PAYMENT TO PRINCIPAL LOAN

This type of loan is structured such that the principal is repaid in equal installments with the interest added to each payment. Therefore, each periodic payment is different; it has a constant amount applied to the principal and a decreasing amount to the interest.

The first part of the program displays the payment number and calculates the payment to interest, total payment, remaining balance, and total interest. The constant payment to principal required as input data (CPMT) can be found by dividing the loan amount by the total number of payment periods. The schedule may be started at any desired payment period; that is, the value entered for K need not be 1.

The second part of the program calculates the accumulated interest between any two payments J and K. The necessary inputs are the periodic interest rate, constant payment, initial loan amount, and the numbers of the starting and ending payments in the time frame.

Equations:

$$BAL_K = PV - (K \times CPMT)$$

$$\text{Kth payment to interest} = (i)\ (BAL_{K-1}) = (PMT_i)_K$$

$$\text{Kth total payment} = CPMT + (PMT_i)_K$$

$$\text{Total interest to payment K} =$$

$$\left[\frac{\frac{(2 - K)\ CPMT}{PV} + 2}{2} \right] [(K - 1)\ (I/100)\ (PV)]$$

Example:

A twenty year 8% loan for $100,000.00 is being amortized by annual payments to principal of $5000.00 plus interest on the remaining balance. Generate a two year amortization schedule on this loan. How much interest is accumulated during years 5 to 10 inclusive?

Solution: (Keystrokes reflect a printer in the system)

Keystrokes:	Display:
[USER]	(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] CPMT [ALPHA]	
[A]	K?
1 [R/S]	INT?
8 [R/S]	CPMT?
5000 [R/S]	PV?
100000 [R/S]	K=1.00
	PMT. I.=8,000.00
	T. PMT.=13,000.00
	BAL.=95,000.00
	T. INT.=8,000.00
	K=2.00
	PMT. I.=7,600.00
	T. PMT.=12,600.00
	BAL.=90,000.00
	T. INT.=15,600.00
[R/S] [B]	INT?
8 [R/S]	CPMT?
5000 [R/S]	PV?
100000 [R/S]	B. PER. NO.?
5 [R/S]	E. PER. NO.?
10 [R/S]	ACC. INT.=32,400.00

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program and set USER mode		USER	
2.	Initialize		[XEQ] CPMT	
3.	To generate an amortization schedule,press		[A]	K ?
4.	Input: first period of sched.(need not			
	be 1)	K	[R/S]	INT ?
	periodic interest rate (%)	INT	[R/S]	CPMT ?
	constant payment to principal	CPMT	[R/S]	PV ?
	initial loan amount	VP	[R/S]	K = 1
5.	Find: payment to interest		[R/S]*	PMT. I.=( )
	total payment		[R/S]*	T. PMT.=( )
	remaining balance		[R/S]*	BAL.=( )
	total interest		[R/S]*	T. INT.=( )
6.	For the next period, press [R/S]		[R/S]*	K = (K+1)
	and go to step 5.			
7.	To find the accumulated interest between			
	any two points, press		[B]	INT ?
8.	Input: periodic interest rate (%)	INT	[R/S]	CPMT ?
	constant payment to principal	CPMT	[R/S]	PV ?
	initial loan amount	PV	[R/S]	B. PER. NO. ?
	beginning period number	J	[R/S]	E. PER. NO. ?
	ending period number	K	[R/S]	ACC. INT.=( )
*	These steps need not be performed when			
	there is a printer in the system.			

Program Listings

```
 01*LBL "CPM
T"
 02 STOP
 03*LBL A
 04 CF 01
 05 "K ?"
 06 PROMPT
 07 STO 00
 08*LBL 01
 09 SF 21
 10 "INT ?"
 11 PROMPT
 12 1 E2
 13 /
 14 STO 01
 15 "CPMT ?"
 16 PROMPT
 17 STO 02
 18 "PV ?"
 19 PROMPT
 20 STO 03
 21 FS? 01
 22 RTN
 23 RCL 00
 24 "K"
 25 XEQ 09
 26*LBL E
 27 RCL 03
 28 RCL 02
 29 RCL 00
 30 *
 31 -
 32 STO 04
 33 RCL 02
 34 +
 35 RCL 01
 36 *
 37 STO 07
 38 1
 39 ST+ 00
 40 RCL 07
 41 "PMT. I.
"
 42 XEQ 09
 43 RCL 02
 44 +
 45 "T. PMT.
"
 46 XEQ 09
 47 RCL 04
 48 "BAL."
```

Prompt and store data

Calc. Amort. sched.

```
 49 XEQ 09
 50*LBL 00
 51 2
 52 RCL 00
 53 -
 54 RCL 02
 55 *
 56 RCL 03
 57 /
 58 2
 59 +
 60 2
 61 /
 62 RCL 00
 63 1
 64 -
 65 *
 66 RCL 01
 67 *
 68 RCL 03
 69 *
 70 FS? 01
 71 RTN
 72 "T. INT.
"
 73 XEQ 09
 74 RCL 03
 75 RCL 02
 76 /
 77 RCL 00
 78 X>Y?
 79 RTN
 80 ADV
 81 "K"
 82 XEQ 09
 83 GTO E
 84*LBL B
 85 SF 01
 86 XEQ 01
 87 "B. PER.
NO. ?"
 88 PROMPT
 89 STO 06
 90 "E. PER.
NO. ?"
 91 PROMPT
 92 1
 93 +
 94 STO 00
 95 XEQ 00
 96 STO 05
```

loan paid off?

Calc. Accumulated interest

Program Listings

```
 97 RCL 06
 98 STO 00
 99 XEQ 00
100 RCL 05
101 X<>Y
102 -
103 "ACC. IN
T."
104*LBL 09
105 "⊢="
106 ARCL X
107 AVIEW
108 RTN
109 .END.
```

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	K	50	
	i/100		
	CPMT		
	PURBAL		
	used		
05	J	55	
	PMTi		
	used		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 41 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01	-	Calc.Acc.Int.	Calc.Amort.Sched.
21	S	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CONSTANT PAYMENT TO
PRINCIPAL LOAN
PROGRAM REGISTERS NEEDED: 34

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 4)

ROW 2 (5 - 10)

ROW 3 (10 - 15)

ROW 4 (15 - 21)

ROW 5 (22 - 30)

ROW 6 (31 - 41)

ROW 7 (41 - 45)

ROW 8 (45 - 48)

ROW 9 (48 - 57)

ROW 10 (58 - 70)

ROW 11 (70 - 73)

ROW 12 (74 - 83)

ROW 13 (83 - 87)

ROW 14 (87 - 90)

ROW 15 (90 - 90)

ROW 16 (91 - 99)

ROW 17 (100 - 103)

ROW 18 (104 - 109)

RULE OF 78's

This program calculates the unearned interest (rebate) as well as the remaining principal due for a prepaid consumer loan using the rule of 78's.

Equations:

$$REB_K = (N - K)\frac{FC\ (N - K + 1)}{N\ (N + 1)}$$

$$BAL_K = (N - K)\quad PMT - REBATE_K$$

Example:

A $1,000 loan, with a total finance charge of $180.00 is being paid at $39.33 per month for 30 months. What is the unearned interest (rebate) and remaining balance after the 25th regular payment?

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 005	
[XEQ] [ALPHA] RULE [ALPHA]	N ?
30 [R/S]	K ?
25 [R/S]	PMT ?
39.33 [R/S]	FC ?
180 [R/S]	REB.=$5.81
[R/S]	BAL=$190.84

User Instructions

SIZE: 005

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program			
2.	Initialize		[XEQ] RULE	N ?
3.	Input: total no. of monthly payments	N	[R/S]	K ?
	no. of last payment made	K	[R/S]	PMT ?
	monthly payment amount	PMT	[R/S]	FC ?
	total finance charge	FC	[R/S]	
4.	Find unearned interest (rebate)			REB=( )
	remaining balance		[R/S]	BAL=( )

Program Listings

```
01◆LBL "RUL
E"
02 "N ?"
03 PROMPT
04 STO 00
05 "K ?"
06 PROMPT
07 STO 01
08 "PMT ?"
09 PROMPT
10 STO 02
11 "FC ?"
12 PROMPT
13 STO 03
14 RCL 00
15 RCL 01
16 -
17 1
18 +
19 RCL 03
20 *
21 RCL 00
22 X↑2
23 LASTX
24 +
25 /
26 RCL 00
27 RCL 01
28 -
29 *
30 STO 04
31 "REB"
32 XEQ 09
33 RCL 02
34 RCL 00
35 RCL 01
36 -
37 *
38 RCL 04
39 -
40 "BAL"
41◆LBL 09
42 "⊢=$"
43 ARCL X
44 PROMPT
45 RTN
46 .END.
```

Prompt and store data

Calculate rebate

Calculate remaining balance

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	N	50	
	K		
	PMT		
	FC		
	REB		
05		55	
0		60	
5		65	
0		70	
5		75	
0		80	
5		85	
)		90	
5		95	

STATUS

SIZE 005 TOT. REG. 17 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

RULE OF 78'S

PROGRAM REGISTERS NEEDED: 12

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 3)

ROW 2 (4 - 8)

ROW 3 (9 - 17)

ROW 4 (18 - 30)

ROW 5 (31 - 38)

ROW 6 (39 - 44)

ROW 7 (45 - 46)

AMORTIZATION SCHEDULE

This program finds both the total interest paid over a specified number of payment periods and the remaining balance at the end of the last specified period, given the periodic interest rate, periodic payment amount, loan amount, and the beginning and ending payment numbers for the time span being considered. The payments associated with both the beginning (J) and the ending (K) payment periods are included in the calculation.

The program can be used for loans with a balloon payment as well as loans arranged to be fully amortized provided two cautions are observed. First, the balloon payment of the loan must be at the same time as, and in addition to, the last payment. Second, care should be taken not to enter a value for K that is after the last payment since the program has no way of knowing the term of the loan.

An option is available to output the amortization schedule between payments J and K.

Equations:

$$BAL_K = \frac{1}{(1+i)^{-K}} \left[PMT \frac{(1+i)^{-K} - 1}{i} + PV \right]$$

$$INT_{J-K} = BAL_K - BAL_{J-1} + (K - J + 1) \cdot PMT$$

where:

Kth payment to principal = $BAL_{K-1} - BAL_K$

Kth payment to interest = $PMT - (BAL_{K-1} - BAL_K)$

Total payment to interest = $(K) \times (PMT) - (PV - BAL_K)$

Notes:

For loans scheduled to be fully amortized, the remaining balance after the last payment period may be slightly more or less than zero. This is because the program assumes that all payments are equal to the value entered for PMT. In fact for most loans, the last payment is slightly more or less than the rest.

The calculator performs all internal calculations to ten digits. If the user wishes to round the schedule to dollars and cents, the following sequence may be used:

1. Press [///] [GTO] · 120
2. [PRGM]
3. [XEQ] [ALPHA] RND [ALPHA]
4. [PRGM]

Example 1:

A mortgage is arranged such that the first payment is made at the end of October, 1978 (i.e., October is payment period 1). It is a $20,000 loan at 9%, with monthly payments of $167.84. What is the accumulated interest for 1978 (periods 1-3) and 1979 (periods 4-15) and what would the remaining balance be at the end of each year?

Keystrokes:	Display:
[USER]	(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 009	
[XEQ] [ALPHA] AMORT [ALPHA]	INT ?
9 [ENTER ↑] 12 [÷] [R/S]	PMT ?
167.84 [R/S]	PV ?
20000 [R/S]	J ?
1 [R/S]	K ?
3 [R/S] [A]	INT=449.60
	BAL=19,946.08
[C]	J ?
4 [R/S]	K ?
15 [R/S] [A]	INT=1,785.89
	BAL=19,717.88

Example 2:

Generate an amortization schedule for the first two payments of a $30,000, 7% mortgage having monthly payments of $200. Then jump ahead and generate the data for the 36th payment.

Solution: (Keystrokes reflect a printer in the system)

[XEQ] [ALPHA] AMORT [ALPHA]	INT ?
7 [ENTER ↑] 12 [÷] [R/S]	PMT ?
200 [R/S]	PV ?
30000 [R/S]	J ?
1 [R/S]	K ?
2 [R/S] [B]	PMT NO.=1.00
	INT=175.00
	PRIN=25.00
	BAL=29,975.00
	ΣINT=175.00
	PMT NO.=2.00

	INT=174.85
	PRIN=25.15
	BAL=29,949.85
	ΣINT=349.85
[C]	J ?
36 [R/S]	K ?
36 [R/S] [B]	PMT NO.=36.00
	INT=169.36
	PRIN=30.64
	BAL=29,001.75
	ΣINT=6,201.75

User Instructions

SIZE: 009

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program and set USER mode		[USER]	
2.	Initialize the program		[XEQ] AMORT	INT ?
3.	Input: periodic interest rate (%)	INT	[R/S]	PMT ?
	periodic payment amount	PMT	[R/S]	PV ?
	initial loan amount	PV	[R/S]	J ?
	starting period no.	J	[R/S]	K ?
	ending period no.	K	[R/S]	
4.	Find the total interest paid between			
	periods J and K inclusive and the balance		[A]	INT= ( )
	at the end of period K - OR -		[R/S]*	BAL=( )
5.	Generate the amortization schedule (J to K)		[B]	PMT NO.=( )
6.			[R/S]*	INT=( )
			[R/S]*	PRIN=( )
			[R/S]*	BAL=( )
			[R/S]*	Σ INT=( )
7.	Press		[R/S]*	PMT NO.=( )
	and go to step 6.			
8.	To change J and K, press		[C]	J ?
	and input: J	J	[R/S]	K ?
	and K	K	[R/S]	
9.	Go to step 4 or 5			
*	These keystrokes need not be performed			
	when there is a printer in the system.			

Program Listings

```
 01◆LBL "AMO
RT"
 02 SF 21
 03 "INT ?"
 04 PROMPT
 05 1 E2
 06 /
 07 STO 01
 08 "PMT ?"
 09 PROMPT
 10 STO 02
 11 "PV ?"
 12 PROMPT
 13 STO 03
 14◆LBL C
 15 "J ?"
 16 PROMPT
 17 STO 07
 18 "K ?"
 19 PROMPT
 20 STO 00
 21 STOP
 22◆LBL A
 23 RCL 00
 24 RCL 07
 25 X<=Y?
 26 GTO 00
 27 STO 00
 28 RDN
 29 STO 07
 30◆LBL 00
 31 1
 32 RCL 01
 33 +
 34 STO 08
 35 RCL 00
 36 XEQ 01
 37 STO 04
 38 RCL 08
 39 RCL 07
 40 1
 41 -
 42 XEQ 01
 43 CHS
 44 RCL 04
 45 +
 46 STO 06
 47 RCL 00
 48 RCL 07
 49 -
 50 1
```

Initialize

Prompt and store data

Calculate total interest between two periods and balance at end

```
 51 +
 52 RCL 02
 53 *
 54 +
 55 "INT"
 56 XEQ 09
 57 RCL 04
 58 "BAL"
 59 XEQ 09
 60 STOP
 61◆LBL B
 62 RCL 07
 63 ADV
 64 "PMT NO.
"
 65 XEQ 09
 66 1
 67 RCL 01
 68 +
 69 STO 08
 70 RCL 07
 71 XEQ 01
 72 STO 04
 73 RCL 08
 74 RCL 07
 75 1
 76 -
 77 XEQ 01
 78 RCL 04
 79 -
 80 STO 06
 81 RCL 02
 82 X<>Y
 83 -
 84 "INT"
 85 XEQ 09
 86 RCL 06
 87 "PRIN"
 88 XEQ 09
 89 RCL 04
 90 "BAL"
 91 XEQ 09
 92 RCL 07
 93 RCL 02
 94 *
 95 RCL 03
 96 RCL 04
 97 -
 98 -
 99 "Σ INT"
100 XEQ 09
```

Generate Amortization schedule

J ≤ K

Program Listings

```
101  1
102  ST+ 07
103  RCL 00
104  RCL 07
105  X<=Y?
106  GTO B
107◆LBL 01
108  CHS
109  Y↑X
110  STO 05
111  1
112  -
113  RCL 01
114  /
115  RCL 02
116  *
117  RCL 03
118  +
119  RCL 05
120  /
121  RTN
122◆LBL 09
123  "⊢="
124  ARCL X
125  AVIEW
126  RTN
127  END
```

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

0	K	50	
	i/100		
	PMT		
	PV		
	used		
5	used	55	
	used		
	J		
	1 + 1/100		
0		60	
5		65	
0		70	
5		75	
0		80	
5		85	
0		90	
5		95	

STATUS

SIZE 009 TOT. REG. 40 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
21	S	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

AMORTIZATION SCHEDULE

PROGRAM REGISTERS NEEDED: 32

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 3)

ROW 2 (3 - 8)

ROW 3 (8 - 14)

ROW 4 (14 - 20)

ROW 5 (21 - 31)

ROW 6 (32 - 42)

ROW 7 (42 - 53)

ROW 8 (54 - 58)

ROW 9 (59 - 64)

ROW 10 (64 - 71)

ROW 11 (71 - 81)

ROW 12 (82 - 87)

ROW 13 (87 - 91)

ROW 14 (92 - 99)

ROW 15 (100 - 107)

ROW 16 (108 - 120)

ROW 17 (121 - 127)

ADD-ON TO APR WITH ODD DAYS



This program calculates the monthly payment amount, total finance charge, and the Annual Percentage Rate (APR) for an add-on rate loan.

When a loan is initiated in the middle of a month, the first payment is generally not required until the end of the first full month. The number of days from the beginning of the loan to the beginning of the first month (see above diagram) are called "odd days" and decrease the APR to be quoted with the loan. The calculation of the APR considers these odd days.

Equations:

$$FC = AMT \bullet \left(\frac{N + h}{12} \right) \bullet AIR$$

$$PMT = \frac{AMT + FC}{N} = AMT \ (1+i)^{h} \left[\frac{i}{1 - (1 + i)^{-N}} \right]$$

$$APR = 12i$$

where:

$$h = ODD \bullet 12/365$$

Example:

A 36 month car loan for $3,500 with a 6% add-on rate is initiated such that there are 18 "odd days". Calculate the monthly payment required to amortize this loan, the total finance charge, and the annual percentage rate.

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] ADD [ALPHA]	ODD ?
18 [R/S]	N ?
36 [R/S]	AIR ?
6 [R/S]	PV ?
3500 [R/S]	PMT=115.01
[R/S]	FC=640.36
[R/S]	APR=10.89

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program			
2.	Initialize the program		[XEQ] ADD	ODD ?
3.	Input: "odd-days" to beginning of month	ODD	[R/S]	N ?
	no. of monthly payments	N	[R/S]	AIR ?
	annual add-on interest rate(%)	AIR	[R/S]	PV ?
	and loan amount	PV	[R/S]	
4.	Find: monthly payment			PMT= ( )
	total finance charge		[R/S]	FC= ( )
	and annual percentage rate		[R/S]	APR= ( )

Program Listings

```
 01*LBL "ADD
"
 02 "ODD ?"
 03 PROMPT
 04 STO 00
 05 12
 06 *
 07 365
 08 /
 09 STO 01
 10 "N ?"
 11 PROMPT
 12 STO 02
 13 "AIR ?"
 14 PROMPT
 15 STO 03
 16 "PV ?"
 17 PROMPT
 18 STO 04
 19 RCL 02
 20 RCL 01
 21 +
 22 12
 23 /
 24 RCL 03
 25 *
 26  E2
 27 /
 28 RCL 04
 29 *
 30 STO 00
 31 RCL 04
 32 +
 33 RCL 02
 34 /
 35 STO 05
 36 "PMT"
 37 XEQ 09
 38 RCL 00
 39 "FC"
 40 XEQ 09
 41 RCL 03
 42 12 E2
 43 /
 44 X=0?
 45 GTO 08
 46 STO 00
 47*LBL 01
 48 1
 49 RCL 00
 50 1
```

Prompt and store data

Calculate payment and finance charge

Calculate f(i)

```
 51 +
 52 STO 06
 53 RCL 02
 54 CHS
 55 Y↑X
 56 STO 07
 57 -
 58 RCL 00
 59 /
 60 RCL 05
 61 *
 62 RCL 06
 63 RCL 01
 64 Y↑X
 65 RCL 04
 66 *
 67 -
 68 RCL 07
 69 RCL 06
 70 /
 71 RCL 02
 72 1
 73 +
 74 *
 75 RCL 00
 76 *
 77 1
 78 RCL 07
 79 -
 80 RCL 00
 81 +
 82 -
 83 RCL 00
 84 X↑2
 85 /
 86 RCL 05
 87 *
 88 RCL 06
 89 RCL 01
 90 Y↑X
 91 RCL 06
 92 /
 93 RCL 01
 94 X<>Y
 95 *
 96 LASTX
 97 -
 98 RCL 04
 99 *
100 -
```

Calculate f'(i)

Program Listings

```
101 /
102 RCL 00
103 X<>Y
104 -
105 STO 00
106 LASTX
107 ABS
108  E6-
109 X<=Y?
110 GTO 01
111 RCL 00
112 1200
113 *
114◆LBL 08
115 "APR"
116◆LBL 09
117 "⊢="
118 ARCL X
119 PROMPT
120 RTN
121 .END.
```

$i_K = i_{K-1} - \frac{f(i)}{f'(i)}$

Display routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	ODD	50	
	h		
	N		
	AIR		
	PV		
05	PMT	55	
	1 + I/100		
	$(1+I/100)^{-N}$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 34 USER MODE
ENG ___ FIX ___ SCI ___ ON ___ OFF X
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ADD-ON TO APR WITH ODD DAYS

PROGRAM REGISTERS NEEDED: 26

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 2)

ROW 2 (3 - 10)

ROW 3 (10 - 16)

ROW 4 (16 - 25)

ROW 5 (26 - 36)

ROW 6 (36 - 41)

ROW 7 (42 - 50)

ROW 8 (51 - 63)

ROW 9 (64 - 76)

ROW 10 (77 - 89)

ROW 11 (90 - 102)

ROW 12 (103 - 112)

ROW 13 (112 - 117)

ROW 14 (118 - 121)

SAVINGS PLAN

This program determines interest earned on a savings account using as input the date and amount of each transaction in the period. Accomodates: a) Periodic or continuous compounding; b) 360 or 365 day convention; c) interest earned or forfeited on withdrawal date; and d) adjusts for leap years. One memory module will be required.

Equations:

For continuous compounding:

$$r = e^{iy/z} - 1$$

For periodic compounding:

$$r = (1+i/n)^{ny/z} - 1$$

$$\text{Interest} = [(1+r)^{d/y} - 1]\, A$$

r = effective annual interest rate
e = constant = 2.718281828 (decimal)
i = nominal annual interest rate (decimal)
y = # days in full year
z = 360 or 365 (interest convention)
n = # of compounding periods per year
d = days of interest
A = Amount of transaction

NOTE:

If the effective annual interest rate is known, rather than the nominal rate, it should be used at step 3.

References: HP-65 USERS' LIBRARY program #02063A by Keith Rumbel
HP-67/HP-97 USERS' LIBRARY program #00288D by Howard Kutner

Example:

Nominal Interest Rate - 5¼%
Continuously compounded
Leap year
Interest on withdrawal date
360 Day basis

Transaction	Date	Amt.
Opening balance	1/1	4377.53
Withdrawal	1/15	700.00
Deposit	3/5	425.00

Solution:

Keystrokes:	Display:
[USER]	(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 012	
[XEQ] [ALPHA] SAVE [ALPHA]	INT ?
5.25 [R/S]	CONT/PER ?
[ALPHA] CONT [ALPHA] [R/S]	LEAP/NORM ?
[ALPHA] LEAP [ALPHA] [R/S]	INT ON W/D DATE ? (Y/N)
[ALPHA] Y [ALPHA] [R/S]	INT BASIS ? (360/365)
360 [R/S]	QUARTER NO. ?
1 [R/S] [A]	DATE (MM.DD) ?
1.01 [R/S]	DEP. AMT ?
4377.53 [R/S]	NEXT TRANS. ?
[B]	DATE (MM.DD) ?
1.15 [R/S]	W/D AMT ?
700 [R/S]	NEXT TRANS. ?
[A]	DATE (MM.DD) ?
3.05 [R/S]	DEP. AMT ?
425 [R/S] [D]	ACC. INT=$52.36
[R/S]	BAL=$4,102.53
[R/S]	T. BAL=$4,154.89

User Instructions

SIZE: 012

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program and set USER mode		[USER]	
2.	Initialize the program		[XEQ] SAVE	INT ?
3.	Input: nominal interest rate (%)	INT	[R/S]	CONT/PER ?
	continuous or periodic compounding	"CONT" or "PER"	[R/S]	LEAP/NORM ?
	leap year or normal year	"LEAP" or "NORM"	[R/S]	INT ON W/D DATE ? (Y/N)
	interest earned on withdrawal date	"Y"or"N"	[R/S]	INT BASIS ? (360/365)
	interest basis	360 or 365	[R/S]	QUARTER NO. ?
	and quarter number of year	1,2,3 or 4	[R/S]	
	TRANSACTIONS:			
4.	For a deposit , press		[A]	DATE(MM.DD)
	input date	MM.DD	[R/S]	DEP. AMT ?
	and amount of deposit	$	[R/S]	NEXT TRANS.?
5.	For a withdrawal, press		[B]	DATE(MM.DD)
	input date	MM.DD	[R/S]	W/D AMT ?
	and amount of withdrawal	$	[R/S]	NEXT TRANS.?
	AT ANY TIME			
6.	Display: accumulated interest		[D]	ACC. INT=$( )
	balance (without interest)		[R/S]	BAL=$( )
	and total balance		[R/S]	T. BAL=$( )
7.	For a new case:			
	a) same parameters	0	[STO] 06	
	(clear accumulating registers)		[STO] 07	
	and go to step 4			
	b) entirely new case, go to step 2			

Program Listings

```
 01◆LBL "SAV
E"
 02 "INT ?"
 03 PROMPT
 04  E2
 05 /
 06 STO 00
 07 0
 08 STO 08
 09 +
 10 CF 00
 11 CF 01
 12 CF 03
 13 CF 02
 14 "CONT/PE
R ?"
 15 PROMPT
 16 ASTO Y
 17 "CONT"
 18 ASTO X
 19 X=Y?
 20 GTO 09
 21 "NO. PER
IODS ?"
 22 PROMPT
 23 STO 08
 24 SF 02
 25◆LBL 09
 26 365
 27 STO 09
 28 CF 00
 29 "LEAP/NO
RM ?"
 30 PROMPT
 31 ASTO Y
 32 "NORM"
 33 ASTO X
 34 X=Y?
 35 GTO 10
 36 SF 00
 37 366
 38 STO 09
 39◆LBL 10
 40 CF 01
 41 "  INT O
N W/D DA"
 42 "├TE ? <
Y/N>"
 43 PROMPT
 44 ASTO Y
 45 "Y"
```

Prompt and store data

```
 46 ASTO X
 47 X=Y?
 48 SF 01
 49 "   INT
BASIS ? "
 50 "├<360/3
65>"
 51 PROMPT
 52 STO 10
 53 "QUARTER
NO. ?"
 54 PROMPT
 55 31
 56 STO 01
 57 STO 02
 58 STO 03
 59 CLX
 60 STO 06
 61 STO 07
 62 +
 63 STO 11
 64 3
 65 *
 66 STO 04
 67 1
 68 DSE 11
 69 GTO 02
 70 RCL 02
 71 3
 72 -
 73 FS? 00
 74 1
 75 FS? 00
 76 +
 77 STO 02
 78 GTO 07
 79◆LBL 02
 80 DSE 11
 81 GTO 03
 82 ST- 01
 83 ST- 03
 84 GTO 07
 85◆LBL 03
 86 DSE 11
 87 GTO 04
 88 ST- 03
 89 GTO 07
 90◆LBL 04
 91 ST- 02
 92◆LBL 07
 93 RCL 09
```

Store No. of days in each month of the quarter

Program Listings

```
 94 RCL 10
 95 /
 96 FS?C 02
 97 GTO 08
 98 RCL 00
 99 *
100 1
101 E↑X
102 X<>Y
103 Y↑X
104 STO 05
105 RTN
106◆LBL 08
107 RCL 08
108 *
109 RCL 00
110 LASTX
111 /
112 1
113 +
114 X<>Y
115 Y↑X
116 STO 05
117 RTN
118◆LBL B
119 XEQ 13
120 "W/D AMT
 ?"
121 PROMPT
122 CHS
123 FS? 01
124 SF 02
125 SF 03
126◆LBL A
127 FC? 03
128 XEQ 13
129 "DEP. AM
T ?"
130 FC?C 03
131 PROMPT
132 ST+ 07
133 X<>Y
134 FS?C 02
135 GTO 03
136 1
137 +
138◆LBL 03
139 RCL 09
140 /
141 RCL 05
142 X<>Y
```

Continuous compounding effective rate

Periodic compounding effective rate

Withdrawal routine

Deposit routine

Interest calculation

```
143 Y↑X
144 1
145 -
146 *
147 RND
148 ST+ 06
149 "NEXT TR
ANS. ?"
150 PROMPT
151◆LBL 13
152 "DATE <M
M.DD>?"
153 PROMPT
154 FRC
155 RCL 04
156 1
157 LASTX
158 INT
159 -
160 +
161 STO 11
162 RCL 03
163 R↑
164  E2
165 *
166 -
167 DSE 11
168 GTO 01
169 RTN
170◆LBL 01
171 RCL 02
172 +
173 DSE 11
174 GTO 02
175 RTN
176◆LBL 02
177 RCL 01
178 +
179 RTN
180◆LBL D
181 RCL 06
182 "ACC. IN
T"
183 XEQ 12
184 RCL 07
185 "BAL"
186 XEQ 12
187 +
188 "T. BAL"
189◆LBL 12
190 "⊢=$"
```

Determine no. of days

Display results

Program Listings

```
191 ARCL X
192 PROMPT
193 RTN
194 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	INT	50	
	#days in month 1		
	#days in month 2		
	#days in month 3		
	3 times quarter#		
05	1+effective int rate	55	
	accumulated int		
	accumulated balance		
	#periods/yr		
	#days/yr		
10	360/365 day basis	60	
	pointer		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 012 TOT. REG. 75 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		leap year	normal year
01		int on W/D date	no int on W/D dat
02		periodic interest	continuous intere
03		W/D transaction	deposit transacti

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SAVINGS PLAN

PROGRAM REGISTERS NEEDED: 64

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 2)

ROW 2 (2 - 11)

ROW 3 (12 - 14)

ROW 4 (14 - 19)

ROW 5 (20 - 21)

ROW 6 (21 - 28)

ROW 7 (28 - 29)

ROW 8 (30 - 35)

ROW 9 (36 - 41)

ROW 10 (41 - 42)

ROW 11 (42 - 44)

ROW 12 (44 - 49)

ROW 13 (49 - 50)

ROW 14 (50 - 53)

ROW 15 (53 - 54)

ROW 16 (55 - 66)

ROW 17 (67 - 75)

ROW 18 (76 - 83)

SAVINGS PLAN

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 19 (84 - 91)

ROW 20 (91 - 101)

ROW 21 (102 - 114)

ROW 22 (115 - 120)

ROW 23 (120 - 125)

ROW 24 (126 - 129)

ROW 25 (129 - 134)

ROW 26 (135 - 146)

ROW 27 (147 - 149)

ROW 28 (149 - 152)

ROW 29 (152 - 158)

ROW 30 (159 - 168)

ROW 31 (169 - 179)

ROW 32 (180 - 183)

ROW 33 (183 - 188)

ROW 34 (188 - 192)

ROW 35 (193 - 194)

INTEREST CONVERSIONS

The first part of the program permits the user to solve for any variable of an accrued simple interest calculation. Given three of the four variables (number of days, annual interest rate, beginning amount, and accrued interest) the fourth is calculated. Accrued interest can be based on a 360 or 365 day year. In addition, the user may choose to add the calculated accrued interest to the initial principal to determine the final amount.

The second part deals with nominal to effective interest rate conversions, and vice-versa. By definition, an annual effective interest rate demonstrates the effect of compounding for a full year of compounding periods at a particular periodic interest rate. The periodic interest rate to be used is determined by dividing the number of compounding periods in a year into the stated annual nominal interest rate. The effect is such that if the nominal rate is held constant, as the number of compounding periods per year is increased, the annual effective interest rate will increase. The ultimate, or upper limit, in this process is to have an infinite number of compounding periods in a year, commonly called continuous compounding.

Given the number of compounding periods in a year, and one of the rates (nominal or effective), the other rate can be calculated. If for example, you require the periodic interest rate for a calculation, given the effective rate, use this program to determine the annual nominal rate first. Dividing the annual nominal rate by the number of compounding periods in a year will give the required periodic interest rate.

The third part is for continuous compounding. Given either rate, the other is calculated.

The most common and straightforward definition of effective interest rate has been implemented. Occasionally other definitions will be used and the results will not compare exactly with those calculated by these programs. For example, since the maximum annual nominal rate that savings institutions can offer is regulated by law, they may modify the process (also regulated) so that the effective rate is even higher (e.g., for daily compounding, the periodic rate may be divided by 360 and then compounding accomplished for 365 periods). It is important then, when attempting to match results, to understand the process employed.

Equations:

$$\text{INT } 360 = \frac{\text{DAYS}}{360} \bullet \text{BEG AMT} \bullet \text{RATE}$$

$$\text{INT } 365 = \frac{\text{DAYS}}{365} \bullet \text{BEG AMT} \bullet \text{RATE}$$

finite coumpounding

$$EFF = \left(1 + \frac{NOM}{C}\right)^{C} - 1$$

continuous compounding

$$EFF = (e^{NOM} - 1)$$

Example 1:

Calculate the accrued interest and final amount (both 360 and 365 day basis) for a $30,000, 8%, 90 day interest at maturity note.

Keystrokes:	Display:	
[USER]		(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 007		
[XEQ] [ALPHA] CONV [ALPHA]		
[A]	INT BASIS (360/365) ?	
365 [R/S]	NO. DAYS ?	
90 [R/S]	INT RATE ?	
8 [R/S]	BEG. AMT ?	
30000 [R/S]	ACC. INT ?	
[R/S]	INT=591.78	
[+]	30,591.78	(Final Amount)

Example 2:

What is the nominal rate if the effective annual rate is 13% compounded quarterly?

Keystrokes:	Display:
[B]	NO. PER. ?
4 [R/S]	NOM ?
[R/S]	EFF ?
13 [R/S]	NOM=12.41

Example 3:

A bank offers a savings plan with a 5% annual nominal interest rate. What is the annual effective rate if compounding is continuous?

Keystrokes:	Display:
[C]	NOM ?
5 [R/S]	EFF ?
[R/S]	C.EFF=5.13

Example 4:

In the above example, what is the annual effective rate if compounding is continuous on a 365/360 basis?

Keystrokes:	Display:
[D]	NOM ?
5 [R/S]	C.EFF=5.20

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program and set USER mode		[USER]	
2.	Initialize		[XEQ] CONV	
3.	SIMPLE INTEREST, press		[A]	INT BASIS (360/365) ?
4.	Input interest basis	360 or 365	[R/S]	NO. DAYS ?
5.	Input 3 of the following:			
	number of days	# days	[R/S]	INT RATE ?
	annual interest rate	INT	[R/S]	BEG. AMT ?
	beginning amount	BEG. AMT	[R/S]	ACC. INT ?
	accrued interest	ACC.INT	[R/S]	
6.	When prompted for the unknown variable,			DAYS=( )
	press [R/S] (make no input). The unknown			or RATE=( )
	is automatically calculated when all the			or AMT=( )
	data is input.			or INT=( )
7.	(Optional) After solving for accrued			
	interest, press		[+]	XXX.XX
	to find the final amount.			
8.	Interest conversions (finite), press		[B]	NO. PER. ?
9.	Input the number of compounding periods/			
	year	NO. PER	[R/S]	NOM ?
10.	Input either one: nominal rate	NOM	[R/S]	EFF ?
	effective rate	EFF	[R/S]	NOM=( )
11.	(See step 6)			or EFF=( )
12.	Interest conversions (continuous), press		[C]	NOM ?
	Input either one: nominal rate	NOM	[R/S]	EFF ?
	effective rate	EFF	[R/S]	C.NOM=( )
13.	(See step 6)			or C.EFF=( )
14.	Calculate the continuous effective rate			

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	(365/360 basis)		[D]	NOM ?
15.	Input nominal rate	NOM	[R/S]	C.EFF=( )

Program Listings

```
 01◆LBL "CON
V"
 02 STOP
 03◆LBL A
 04 1.1
 05 STO 00
 06 "INT BAS
IS <360/"
 07 "÷365> ?
"
 08 PROMPT
 09 STO 05
 10 CF 22
 11 "NO. DAY
S ?"
 12 XEQ 12
 13 "INT RAT
E ?"
 14 XEQ 12
 15 "BEG. AM
T ?"
 16 XEQ 12
 17 "ACC. IN
T ?"
 18 XEQ 12
 19 1 E2
 20 ST/ 02
 21 GTO IND
06
 22◆LBL 01
 23 RCL 04
 24 RCL 05
 25 *
 26 RCL 03
 27 /
 28 RCL 02
 29 /
 30 "DAYS"
 31 XEQ 13
 32◆LBL 02
 33 RCL 05
 34 RCL 04
 35 *
 36 RCL 01
 37 /
 38 RCL 03
 39 /
 40 1 E2
 41 *
 42 "RATE"
 43 XEQ 13
```

Simple interest

Prompt and store data

Calculate no. of days

Calculate Int. rate.

```
 44◆LBL 03
 45 RCL 04
 46 RCL 05
 47 *
 48 RCL 01
 49 /
 50 RCL 02
 51 /
 52 "AMT"
 53 XEQ 13
 54◆LBL 04
 55 RCL 03
 56 RCL 01
 57 RCL 05
 58 /
 59 RCL 03
 60 *
 61 RCL 02
 62 *
 63 "INT"
 64 XEQ 13
 65◆LBL B
 66 1.1
 67 STO 00
 68 "NO. PER
. ?"
 69 PROMPT
 70 STO 05
 71◆LBL 14
 72 CF 22
 73 "NOM ?"
 74 XEQ 12
 75 "EFF ?"
 76 XEQ 12
 77 GTO IND
06
 78◆LBL 01
 79 RCL 02
 80 1 E2
 81 /
 82 1
 83 +
 84 RCL 05
 85 1/X
 86 Y↑X
 87 1
 88 -
 89 RCL 05
 90 *
 91 1 E2
 92 *
```

Calculate beginning amount

Calculate accumulated interest

Nom. (-) eff.

Prompt and store data

Calculate nominal rate

Program Listings

```
 93 "NOM"
 94 XEQ 13
 95◆LBL 02
 96 RCL 01
 97 RCL 05
 98 1 E2
 99 *
100 /
101 1
102 +
103 RCL 05
104 Y↑X
105 1
106 -
107 1 E2
108 *
109 "EFF"
110 XEQ 13
111◆LBL C
112 3.1
113 STO 00
114 GTO 14
115◆LBL 03
116 RCL 04
117 1 E2
118 /
119 1
120 +
121 LN
122 1 E2
123 *
124 "C.NOM"
125 XEQ 13
126◆LBL 04
127 RCL 03
128◆LBL 07
129 1 E2
130 /
131 E↑X
132 1
133 -
134 1 E2
135 *
136 "C.EFF"
137◆LBL 13
138 "⊢="
139 ARCL X
140 PROMPT
141 RTN
142◆LBL 12
143 PROMPT
144 STO IND
00
145 RCL 00
146 FC?C 22
147 STO 06
148 ISG 00
149 RTN
150◆LBL D
151 "NOM ?"
152 PROMPT
153 365
154 *
155 360
156 /
157 GTO 07
158 .END.
```

Calculate effective rate

Continuous compounding

Calculate nominal rate

Calculate effective rate

Display routine

Input routine

input made? no, calc. this quantity

Continuous 365/360 basis

70

80

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	pointer	50	
	No. days/NOM		
	Int rate/EFF		
	B. AMT/CNOM		
	ACC. INT/CEFF		
05	360 or 365/no. per	55	
	subroutine pointer		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 57 USER MODE
ENG ___ FIX 2 SCI ___ ON X OFF ___
DEG -- RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
22	C	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

INTEREST CONVERSIONS

PROGRAM REGISTERS NEEDED: 51

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 4)

ROW 2 (4 - 6)

ROW 3 (6 - 7)

ROW 4 (8 - 11)

ROW 5 (11 - 13)

ROW 6 (13 - 15)

ROW 7 (15 - 17)

ROW 8 (17 - 20)

ROW 9 (21 - 30)

ROW 10 (30 - 39)

ROW 11 (40 - 44)

ROW 12 (45 - 53)

ROW 13 (53 - 63)

ROW 14 (63 - 68)

ROW 15 (68 - 72)

ROW 16 (73 - 75)

ROW 17 (75 - 81)

ROW 18 (82 - 92)

ROW 19 (93 - 98)

ROW 20 (99 - 109)

ROW 21 (109 - 114)

ROW 22 (114 - 122)

ROW 23 (123 - 128)

ROW 24 (129 - 136)

ROW 25 (136 - 142)

ROW 26 (143 - 151)

ROW 27 (151 - 155)

ROW 28 (156 - 158)

LEASE WITH ADDITIONAL PAYMENTS IN ADVANCE

Payments on loans are typically made at the end of the period (in arrears). However, there are situations where payments are made in advance (leasing is a good example). Sometimes these agreements call for extra payments to be made when the transaction is closed, before the payments would normally be due. Or, the transaction has advance payments and a residual value at the end of the normal term.

This program solves for the periodic payment amount necessary to achieve a desired yield when a number of payments are made in advance. And, given the periodic payment, the program finds the yield. Either amount may be calculated when a residual value exists.

The necessary inputs are the total number of periods in the loan (n), the number of payments made in advance (A), the loan amount (PV), and either the periodic payment amount (PMT) or the periodic yield (i). The residual value at the end of the nth period (RESID) is optional.

Equations:

$$PMT = \frac{PV - RESID\ (1 + i)^{-n}}{\left[\frac{1 - (1 + i)^{-(n-A)}}{i} + A\right]}$$

Notes:

The value of A must be less than the value of n. A = 0 implies an ordinary annuity calculation, while A = 1 means an annuity due calculation.

Example:

A lease has been written to run for 60 months. The leased equipment has a value of $25,000 with a $600 monthly payment. The lessee has agreed to make 3 payments at the time of closing. What is the annual yield? (There is no residual value at the end of 60 months.)

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] ADV [ALPHA]	NO. ADV. PMTS?
3 [R/S]	NO. PER. ?
60 [R/S]	PV ?
25000 [R/S]	RESID. ?
0 [R/S]	PMT ?
600 [R/S]	INT=1.44
12 [x]	17.33 (annual)

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program			
2.	Initialize the program		[XEQ] ADV	NO. ADV. PMTS
3.	Input: no. of payments made in advance	A	[R/S]	NO. PER. ?
	no. of periods in the of loan	n	[R/S]	PV ?
	loan amount	PV	[R/S]	RESID. ?
	residual value at end of nth period	RESID	[R/S]	PMT ?
	and periodic payment	PMT	[R/S]	INT=( )
	(if unknown, press		[R/S]	INT ?
	and input periodic interest)	INT	[R/S]	PMT=( )

Program Listings

```
01◆LBL "ADV
"
02 "NO. ADV
. PMTS?"
03 PROMPT
04 STO 01
05 "NO. PER
. ?"
06 PROMPT
07 X<=Y?
08 GTO 02
09 CHS
10 STO 00
11 "PV ?"
12 PROMPT
13 STO 04
14 "RESID.
?"
15 PROMPT
16 STO 05
17 CF 22
18 "PMT ?"
19 PROMPT
20 FS?C 22
21 GTO 01
22 "INT ?"
23 PROMPT
24  E2
25 /
26 STO 02
27 1
28 +
29 STO 07
30 RCL 00
31 Y↑X
32 RCL 05
33 *
34 RCL 04
35 X<>Y
36 -
37 RCL 07
38 RCL 00
39 RCL 01
40 +
41 Y↑X
42 1
43 X<>Y
44 -
45 RCL 02
46 /
47 RCL 01
```

Prompt and store data

error

Calculate payment

```
48 +
49 /
50 "PMT"
51 XEQ 12
52◆LBL 01
53 STO 03
54  E-3
55 STO 02
56◆LBL 00
57 1
58 RCL 02
59 +
60 STO 07
61 RCL 00
62 RCL 01
63 +
64 Y↑X
65 1
66 X<>Y
67 -
68 RCL 02
69 /
70 RCL 01
71 +
72 RCL 03
73 *
74 RCL 07
75 RCL 00
76 Y↑X
77 RCL 05
78 *
79 +
80 RCL 04
81 -
82 STO 06
83 RCL 07
84 RCL 00
85 RCL 01
86 +
87 1
88 -
89 Y↑X
90 RCL 00
91 CHS
92 RCL 01
93 -
94 *
95 RCL 02
96 *
97 RCL 07
98 RCL 00
99 RCL 01
```

Calculate interest using Newton method

Program Listings

```
100 +
101 Y↑X
102 1
103 X<>Y
104 -
105 -
106 RCL 02
107 X↑2
108 /
109 RCL 03
110 *
111 RCL 07
112 RCL 00
113 1
114 -
115 Y↑X
116 RCL 05
117 *
118 RCL 00
119 *
120 +
121 RCL 06
122 X<>Y
123 /
124 ST- 02
125 ABS
126  E-6
127 X<=Y?
128 GTO 00
129 RCL 02
130 1 E2
131 *
132 "INT"
133◆LBL 12
134 "⊢="
135 ARCL X
136 PROMPT
137 RTN
138◆LBL 02
139 0
140 /
141 END
```

Display routine

"DATA ERROR"

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	-n	50	
	A		
	i/100		
	PMT		
	PV		
05	RESID	55	
	f(i)		
	1 + i/100		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 40 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
22	C	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

LEASE WITH ADDITIONAL PAYMENTS
IN ADVANCE
PROGRAM REGISTERS NEEDED: 32

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 2)

ROW 2 (2 - 5)

ROW 3 (5 - 8)

ROW 4 (9 - 14)

ROW 5 (14 - 18)

ROW 6 (18 - 22)

ROW 7 (23 - 34)

ROW 8 (35 - 47)

ROW 9 (48 - 54)

ROW 10 (54 - 66)

ROW 11 (67 - 79)

ROW 12 (80 - 92)

ROW 13 (93 - 105)

ROW 14 (106 - 118)

ROW 15 (119 - 128)

ROW 16 (128 - 134)

ROW 17 (134 - 141)

SKIPPED PAYMENTS

Sometimes a loan (or lease) may be negotiated in which a specific set of monthly payments are going to be skipped each year. Seasonality is usually the reason for such an agreement. For example, because of heavy rainfall, a bulldozer cannot be operated in Oregon during December, January, and February, and the lessee wishes to make payments only when his machinery is being used. He will make nine payments per year, but the interest will continue to accumulate over the months in which a payment is not made.

Equations:

$$D_{END} = \frac{E}{\left[1 - \left(1 + \frac{C}{A}\right)^{-AB}\right]} \times \frac{\left[\left(1 + \frac{C}{A}\right)^{A} - 1\right]\frac{C}{A}}{\left[\left(1 + \frac{C}{A}\right)^{A} - \left(1 + \frac{C}{A}\right)^{A-K} + \left(1 + \frac{C}{A}\right)^{A-L-K} - 1\right]}$$

$$D_{BEGIN} = \frac{D_{END}}{1 + \frac{C}{A}}$$

where: A = number of payment periods per year
B = number of years
C = annual percentage rate (as decimal)
D = periodic payment amount
E = loan amount
K = number of last payment before payments close the first time
L = number of skipped payments

Example:

A bulldozer worth $100,000 is being purchased in September. The first payment is due one month later, and payments will continue over a period of 5 years. Due to the weather, the machinery will not be used during the winter months, and the purchaser does not wish to make payments during January, February, and March (months 4 thru 6). If the current interest rate is 8 3/4%, what is the monthly payment necessary to amortize the loan?

Solution:

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] SKIP [ALPHA]	NO. PER./YR. ?
12 [R/S]	NO. YRS. ?
5 [R/S]	INT ?
8.75 [R/S]	PV ?
100000 [R/S]	LAST PMT NO. ?
3 [R/S]	NO. PMTS SKIPPED ?
3 [R/S]	E. PMT=2,761.44

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in program			
2.	Initialize		[XEQ] SKIP	NO. PER./YR. ?
3.	Input: the number of periods/year	NO.PER/YR	[R/S]	NO. YRS. ?
	number of years	NO. YRS	[R/S]	INT ?
	annual interest (%)	INT	[R/S]	PV ?
	loan amount	PV	[R/S]	LAST PMT NO.
	last payment no.	LAST PMT	[R/S]	NO. PMTS SKIPPED ?
	and no. of payments skipped	NO.SKIPPED	[R/S]	
4.	Find: payment (arrears)			E. PMT=C
	and payment (beginning)		[R/S]	B. PMT=C

Program Listings

```
 01◆LBL "SKI
P"
 02 "NO. PER
./YR. ?"
 03 PROMPT
 04 STO 00
 05 "NO. YRS
. ?"
 06 PROMPT
 07 STO 01
 08 "INT ?"
 09 PROMPT
 10 100
 11 /
 12 STO 02
 13 "PV ?"
 14 PROMPT
 15 STO 04
 16 "LAST PM
T NO.?"
 17 PROMPT
 18 STO 05
 19 "   NO.
PMTS SKI"
 20 "├PPED ?
"
 21 PROMPT
 22 STO 06
 23 RCL 04
 24 RCL 02
 25 RCL 00
 26 /
 27 1
 28 +
 29 STO 07
 30 RCL 00
 31 RCL 01
 32 *
 33 CHS
 34 Y↑X
 35 1
 36 -
 37 CHS
 38 /
 39 RCL 07
 40 RCL 00
 41 Y↑X
 42 1
 43 -
 44 STO 03
 45 RCL 02
```

Prompt and store data

Calculate payment

```
 46 *
 47 *
 48 RCL 00
 49 /
 50 RCL 07
 51 RCL 00
 52 RCL 05
 53 -
 54 Y↑X
 55 ST- 03
 56 CLX
 57 RCL 07
 58 RCL 00
 59 RCL 06
 60 -
 61 RCL 05
 62 -
 63 Y↑X
 64 RCL 03
 65 +
 66 /
 67 "E. PMT"
 68 XEQ 12
 69 RCL 07
 70 /
 71 "B. PMT"
 72◆LBL 12
 73 "├="
 74 ARCL X
 75 PROMPT
 76 RTN
 77 .END.
```

Display routine

90	
00	

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	A	50	
	B		
	C		
	used		
	E		
05	K	55	
	L		
	1 + C/A		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 32 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SKIPPED PAYMENTS

PROGRAM REGISTERS NEEDED: 25

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 2)

ROW 2 (2 - 5)

ROW 3 (5 - 8)

ROW 4 (8 - 13)

ROW 5 (13 - 16)

ROW 6 (16 - 19)

ROW 7 (19 - 20)

ROW 8 (20 - 27)

ROW 9 (28 - 40)

ROW 10 (41 - 53)

ROW 11 (54 - 65)

ROW 12 (66 - 70)

ROW 13 (71 - 74)

ROW 14 (75 - 77)

COMPOUNDING PERIODS DIFFERENT FROM PAYMENT PERIODS



Payments into a savings plan may not occur with the same frequency as the compounding frequency offered. This program solves for the number of payments, the periodic payment amount, or future value.

The diagrams above depict two of the many combinations that may be encountered. Note that payments are assumed to occur at the beginning of the payment period (annuity due).

Another assumption of this program is that payments deposited for a partial compounding period will accrue simple interest for the remainder of the compounding period. Thus, a deposit at the beginning of the 2nd month of a quarter into a savings plan that compounds quarterly is assumed to accrue two months simple interest. This is often the case, but is not true for all institutions.

Equations:

$$PMT = \frac{FV}{Z}\left[\frac{Q}{(1+Q)^{n}-1}\right]$$

when $P/C \leq 1$

$Q = (1 + i)^{C/P} - 1$

$n = \#PAY$

$Z = (1 + Q)$

when $P/C > 1$

$Q = i$

$n = (\#PAY) \times (C/P)$

$Z = (P/C + 1) \times (\frac{Q}{2}) + (P/C)$

Example 1:

Quarterly deposits of $95 are to be made into a savings account paying 5% compounded monthly. What amount will be in that account after 7 years (28 total payments)?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] CPDPP [ALPHA]	NO. PMT/YR ?
4 [R/S]	NO. PER./YR ?
12 [R/S]	INT ?
5 [ENTER ↑] 12 [÷] [R/S]	NO. PMTS ?
7 [ENTER ↑] 4 [x] [R/S]	PMT ?
95 [R/S]	FV ?
[R/S]	FV=3,203.59

Example 2:

In 2 years, you will need $4000. If a savings account will pay 5¼% compounded quarterly, what amount must you deposit each month to accumulate the desired amount:

Keystrokes:	Display:
[XEQ] [ALPHA] CPDPP [ALPHA]	NO. PMT/YR ?
12 [R/S]	NO. PER./YR ?
4 [R/S]	INT ?
5.25 [ENTER ↑] 4 [÷] [R/S]	NO. PMTS ?
24 [R/S]	PMT ?
[R/S]	FV ?
4000 [R/S]	PMT=157.78

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program			
2.	Initialize the program		[XEQ] CPDPP	NO. PMT/YR ?
3.	Input: number of payment periods per year	NO. PMT/YR	[R/S]	NO. PER./YR ?
	periodic interest rate (%)	NO. PMT/YR	[R/S]	NO. PMTS ?
4.	Input two of the following:			
	total number of payments	NO. PMTS	[R/S]	PMT ?
	periodic payment amount	PMT	[R/S]	FV ?
	and future value	FV	[R/S]	
5.	When prompted for the unknown quantity,			N=( )
	press [R/S] (make no input). The unknown			or PMT=( )
	is then automatically calculated when all			or FV=( )
	the data is input.			
6.	For a new case, got to step 2			

Program Listings

```
 01◆LBL "CPD
PP"
 02 1
 03 "NO. PMT
/YR ?"
 04 PROMPT
 05 "NO. PER
./YR ?"
 06 PROMPT
 07 /
 08 STO 04
 09 X>Y?
 10 SF 00
 11 "INT ?"
 12 PROMPT
 13 100
 14 /
 15 STO 03
 16 LASTX
 17 *
 18 RCL 03
 19 1
 20 +
 21 RCL 04
 22 1/X
 23 Y↑X
 24 STO 05
 25 .1
 26 STO 06
 27 CF 22
 28 "NO. PMT
S ?"
 29 XEQ 09
 30 "PMT ?"
 31 XEQ 09
 32 "FV ?"
 33 XEQ 09
 34 GTO IND
07
 35◆LBL 00
 36 FS?C 00
 37 GTO 10
 38 RCL 05
 39 1
 40 -
 41 RCL 02
 42 *
 43 RCL 05
 44 RCL 01
 45 *
 46 /
```

Prompt and store data

P/C > 1

Calculate N
P/C ≤ 1

```
 47 1
 48 +
 49 LN
 50 RCL 05
 51 LN
 52 /
 53 "N"
 54 XEQ 13
 55◆LBL 10
 56 RCL 02
 57 RCL 03
 58 *
 59 RCL 04
 60 1
 61 +
 62 RCL 03
 63 2
 64 /
 65 *
 66 RCL 04
 67 +
 68 RCL 01
 69 *
 70 /
 71 1
 72 +
 73 LN
 74 RCL 03
 75 1
 76 +
 77 LN
 78 /
 79 RCL 04
 80 *
 81 "N"
 82 XEQ 13
 83◆LBL 01
 84 FS?C 00
 85 GTO 11
 86 RCL 05
 87 1
 88 -
 89 RCL 05
 90 RCL 00
 91 Y↑X
 92 1
 93 -
 94 /
 95 RCL 02
 96 *
 97 RCL 05
```

Calculate N
P/C > 1

Calculate PMT
P/C ≤ 1

Program Listings

```
 98 /
 99 "PMT"
100 XEQ 13
101◆LBL 11
102 RCL 04
103 1/X
104 RCL 00
105 *
106 RCL 03
107 1
108 +
109 X<>Y
110 Y↑X
111 1
112 -
113 RCL 03
114 X<>Y
115 /
116 RCL 04
117 1
118 +
119 RCL 03
120 2
121 /
122 *
123 RCL 04
124 +
125 /
126 RCL 02
127 *
128 "PMT"
129 XEQ 13
130◆LBL 02
131 FS?C 00
132 GTO 12
133 RCL 05
134 RCL 00
135 Y↑X
136 1
137 -
138 RCL 05
139 *
140 RCL 01
141 *
142 RCL 05
143 1
144 -
145 /
146 "FV"
147 XEQ 13
148◆LBL 12
```

Calculate PMT
P/C > 1

Calculate FV
P/C ≤ 1

```
149 RCL 04
150 1
151 +
152 RCL 03
153 2
154 /
155 *
156 RCL 04
157 +
158 RCL 03
159 1
160 +
161 RCL 00
162 RCL 04
163 1/X
164 *
165 Y↑X
166 1
167 -
168 *
169 RCL 01
170 *
171 RCL 03
172 /
173 "FV"
174◆LBL 13
175 "⊢="
176 ARCL X
177 PROMPT
178 RTN
179◆LBL 09
180 PROMPT
181 STO IND
06
182 RCL 06
183 FC?C 22
184 STO 07
185 ISG 06
186 RTN
187 .END.
```

Calculate FV
P/C > 1

Display routine

Input routine

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	N	50	
	PMT		
	FV		
	i/100		
	P/C		
05	(1+i) C/P	55	
	pointer		
	subroutine pointer		
0		60	
5		65	
0		70	
5		75	
0		80	
5		85	
0		90	
5		95	

STATUS

SIZE 008 TOT. REG. 49 USER MODE

ENG ___ FIX 2 SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	P/C > 1	P/C ≤ 1
22	C	refer to owner's manual	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

COMPOUNDING PERIODS DIFFERENT
FROM PAYMENT PERIODS
PROGRAM REGISTERS NEEDED: 42

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 3)

ROW 2 (3 - 5)

ROW 3 (5 - 6)

ROW 4 (7 - 13)

ROW 5 (13 - 24)

ROW 6 (25 - 28)

ROW 7 (28 - 31)

ROW 8 (31 - 35)

ROW 9 (36 - 46)

ROW 10 (47 - 56)

ROW 11 (57 - 69)

ROW 12 (70 - 81)

ROW 13 (82 - 90)

ROW 14 (91 - 100)

ROW 15 (100 - 111)

ROW 16 (112 - 124)

ROW 17 (125 - 131)

ROW 18 (132 - 143)

COMPOUNDING PERIODS DIFFERENT
FROM PAYMENT PERIODS

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 19 (144 - 152)

ROW 20 (153 - 165)

ROW 21 (166 - 175)

ROW 22 (175 - 184)

ROW 23 (185 - 187)

COMPOUND INTEREST SOLUTIONS

Commonly described as annuities and compound amounts, this program converts your HP-41C into a financial calculator, giving you the ability to solve complex problems involving savings, mortgages, annuities, and other financial calculations in a simple and straightforward manner. It duplicates the convenient and powerful built-in functions of the "top row keys" found on HP financial calculators. One Memory Module is needed to execute the program.

The five variables which have become standard for formatting and describing most compound interest problems can best be explained by referring to a pictorial representation called the cash flow diagram.



The diagram begins with a horizontal line called the time line. It represents the duration of a financial problem and is divided into N compounding periods of equal duration (length).

Exchange of cash is represented with vertical arrows. Money received is represented by an arrow pointing up (positive) from the time line where the transaction occured and money paid out is represented by an arrow pointing down (negative).

Payments (PMT) represent a series of cash exchanges of the same direction and amount. In the standard cash flow diagram the payments occur coincidental with the compounding periods and are equal to the number of periods. The first payment can either occur at the beginning of the first period (BEGIN) or at the end of the first period (END).

It is always necessary when working compound interest problems involving payments (PMT) to specify which of the two possible payment streams is applicable, (BEGIN) or (END). In the parlance of various industries BEGIN payments are often referred to as annuity due, or first payment in advance. END payments are referred to as ordinary annuity, payment in arrears, or immediate annuity.

A single cash flow at the start of the time line is called the present value (PV). A similar single cash flow at the end of the time line is called the future value (FV).

The fifth variable is I, the compound interest rate per period.

This program solves for any of the five standard compound interest variables:

N = the number of payments or compounding periods
I = the interest rate per period (as a percent)
PV = the initial transaction (present value)
PMT = the periodic payment coinciding with the compounding period
FV = the final transaction (future value)

When using the cash flow diagram and the cash flow sign convention to format compound interest problems the following rules always apply.

- N and I must correspond to the same period of time
- Both N and I must be present in a problem. Either both values are known, or one is known and the other is to be solved for.
- A valid financial transaction must always include at least one positive cash flow and one negative cash flow.

The cash flow diagram can be used to describe many variations of compound interest problems. Although the terminology used to describe a particular cash transaction may vary from industry to industry the cash flow diagram remains consistent. In providing a means of describing financial problems without using terminology specific to a particular segment, the cash flow diagram becomes, in a sense, a universal language.

Equations:

$$0 = PV + (1+\delta i)\ PMT \left[\frac{1 - (1+i)^{-N}}{i}\right] + FV\ (1+i)^{-N}$$

where $i = I/100$

$$\delta = \begin{cases} 0 & \text{in END} \\ 1 & \text{in BEGIN} \end{cases}$$

Example 1:

What monthly payment will amortize a mortgage loan of $50,000 over 30 years at 10½% interest? The first payment is made 1 month after the exchange of the initial loan amount (END).

Keystrokes:	Display:	
[USER]		(Set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 010		
[XEQ] [ALPHA] MONEY [ALPHA]	0.00	
30 [///] [A]	N=360.00	
50000 [C]	PV=50,000.00	
10.5 [///] [B]	I=0.88	
[D]	PMT=-457.37	(Monthly payment)

Example 2:

In the previous example, what amount would be necessary to prepay the mortgage (remaining balance) at the end of the 6th year?

Keystrokes:	Display:	
6 [///] [A]	N=72.00	
[E]	FV=-48,018.77	(Remaining balance)

Example 3:

How much money must be set aside in a savings account each month in order to accumulate $4,000 in three years if the account compounds monthly at 6% per year? The deposits "begin" immediately.

Keystrokes:	Display:	
[///] [E]	0.00	(Clears financial data registers)
[///] [C]	BEGIN	(Set BEGIN mode)
4000 [E]	FV=4,000.00	
3 [///] [A]	N=36.00	
6 [///] [B]	I=0.50	
[D]	PMT=-101.18	(Monthly deposit)

Example 4:

What interest rate did the bank pay (in the previous example) if the actual amount at the end of the 3 years was $4,025.50?

Keystrokes:	Display:	
4025.50 [E]	FV=4,025.50	
[B]	I=0.53	
12 [X]	6.40	(Annual interest rate)

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Key in the program and set USER mode.		[USER]	
2.	Initialize		[XEQ] MONEY	0.00
3.	The following steps may be performed in			
	any order:			
	• Multiplies the displayed number by			
	12 and stores in N	n	[///] [A]	N = n x 12
	• Divides the displayed number by 12			
	and stores in I	i	[///] [B]	I = i/12
	• Toggles between BEGIN and END modes.			
	Flag 0 displayed (set) is BEGIN mode.		[///] [C]	BEGIN or END
	• List values*		[///] [D]	
	• Clear financial data		[///] [E]	0.00
4.	The following steps may be performed in			
	any order:			
	• Compute or store number of periods+		[A]	N =
	• Compute or store compound interest			
	rate+		[B]	I =
	• Compute or store present value+		[C]	PV =
	• Compute or store payment+		[D]	PMT =
	• Compute or store future value+		[E]	FV =
5.	Review stored values		[RCL] [A]-[E]	
*	Press [R/S] to list successive values if a printer is not being used.			
+	If an [A]-[E] key is pressed immediately after keying in a value, the value will be stored. If the key is pressed after previously pressing another [A]-[E] key and during which time no digit entry has been made, computation will occur.			

Program Listings

```
 01*LBL "MON
EY"
 02*LBL e
 03 SF 21
 04 SF 27
 05 CF 00
 06 FIX 2
 07 0
 08 STO 00
 09 STO 01
 10 STO 02
 11 STO 03
 12 STO 04
 13 STO 05
 14 RTN
 15*LBL 14
 16 "├="
 17 ARCL X
 18 AVIEW
 19 RTN
 20*LBL a
 21 12
 22 *
 23*LBL A
 24 "N"
 25 STO 01
 26 FS?C 22
 27 GTO 14
 28 XEQ "N"
 29 GTO 14
 30*LBL b
 31 12
 32 /
 33*LBL B
 34 "I"
 35 STO 02
 36 FS?C 22
 37 GTO 14
 38 XEQ "*I"
 39 GTO 14
 40*LBL C
 41 "PV"
 42 STO 03
 43 FS?C 22
 44 GTO 14
 45 XEQ "PV"
 46 GTO 14
 47*LBL D
 48 "PMT"
 49 STO 04
 50 FS?C 22
```

Initialize

Output routine

12

Store N

12÷

Store I

Store PV

Store PMT

```
 51 GTO 14
 52 XEQ "PMT
"
 53 GTO 14
 54*LBL E
 55 "FV"
 56 STO 05
 57 FS?C 22
 58 GTO 14
 59 XEQ "FV"
 60 GTO 14
 61*LBL c
 62 "END"
 63 0
 64 STO 00
 65 FS?C 00
 66 PROMPT
 67 1
 68 STO 00
 69 SF 00
 70 "BEGIN"
 71 PROMPT
 72*LBL d
 73 ADV
 74 FS? 00
 75 GTO 00
 76 "END"
 77 AVIEW
 78 GTO 01
 79*LBL 00
 80 "BEGIN"
 81 AVIEW
 82*LBL 01
 83 "N"
 84 RCL 01
 85 XEQ 14
 86 "I"
 87 RCL 02
 88 XEQ 14
 89 "PV"
 90 RCL 03
 91 XEQ 14
 92 "PMT"
 93 RCL 04
 94 XEQ 14
 95 "FV"
 96 RCL 05
 97 GTO 14
 98*LBL "N"
 99 RCL 02
100 X=0?
```

Store FV

Begin/End

List variables

Calculate N

Program Listings

```
101 GTO 03
102 1 E2
103 /
104 STO 06
105 RCL 03
106 *
107 RCL 04
108 +
109 X=0?
110 GTO 04
111 XEQ 08
112 RCL 09
113 SIGN
114 X>0?
115 GTO 02
116 RCL 08
117 SIGN
118 X>0?
119 GTO 01
120*LBL 04          Data error
121 "$ERROR"        message
122 AVIEW
123 RTN
124*LBL 03
125 RCL 03          Calculate N if
126 RCL 05          I=0
127 +
128 RCL 04
129 /
130 CHS
131 STO 01
132 RTN
133*LBL 01
134 RCL 08
135 XEQ 10
136 CHS
137 STO 01
138 RTN
139*LBL 02
140 RCL 09
141*LBL 10
142 LN1+X
143 RCL 06
144 LN1+X
145 /
146 STO 01
147 RTN
148*LBL 09
149 RCL 03
150 ENTER↑
151 X<> 05
152 STO 03
153 +
154 RCL 06
155 CHS
156 STO 06
157 ABS
158 RCL 00
159 *
160 1
161 +
162 RCL 04
163 *
164 RCL 06
165 RCL 03
166 *
167 +
168 /
169 RCL 06
170 *
171 CHS
172 STO 09
173 RTN
174*LBL 08
175 XEQ 09
176 STO 08
177 XEQ 09
178 RTN
179*LBL "PV"
180 RCL 02          Calculate PV
181 X=0?
182 GTO 00
183 XEQ 07
184 RCL 07
185 RCL 05
186 *
187 RCL 04
188 RCL 06
189 *
190 +
191 GTO 01
192*LBL 00
193 RCL 04          Calculate PV
194 RCL 01          if I=0
195 *
196 RCL 05
197 +
198*LBL 01
199 CHS
200 STO 03
201 RTN
202*LBL "PMT         Calculate PMT
```

Program Listings

```
"
203 RCL 02
204 X=0?
205 GTO 00
206 XEQ 07
207 RCL 07
208 RCL 05
209 *
210 RCL 03
211 +
212 RCL 06
213 /
214 GTO 01
215◆LBL 00
216 RCL 03
217 RCL 05
218 +
219 RCL 01
220 /
221◆LBL 01
222 CHS
223 STO 04
224 RTN
225◆LBL "FV"
226 RCL 02
227 X=0?
228 GTO 00
229 XEQ 07
230 RCL 06
231 RCL 04
232 *
233 RCL 03
234 +
235 RCL 07
236 /
237 GTO 01
238◆LBL 00
239 RCL 04
240 RCL 01
241 *
242 RCL 03
243 +
244◆LBL 01
245 CHS
246 STO 05
247 RTN
248◆LBL 07
249 RCL 02
250 1 E2
251 /
252 LN1+X
```

Calculate PMT if I=0

Calculate FV

Calculate FV if I=0

```
253 RCL 01
254 *
255 CHS
256 E↑X
257 STO 07
258 LASTX
259 E↑X-1
260 CHS
261 RCL 02
262 1 E2
263 /
264 /
265 LASTX
266 RCL 00
267 *
268 1
269 +
270 *
271 STO 06
272 RTN
273◆LBL "*I"
274 RCL 01
275 RCL 04
276 *
277 RCL 03
278 +
279 RCL 05
280 +
281 X=0?
282 GTO 02
283 CF 05
284 RCL 03
285 RCL 04
286 RCL 00
287 *
288 +
289 STO 06
290 LASTX
291 RCL 04
292 -
293 CHS
294 RCL 05
295 +
296 STO 07
297 RCL 01
298 1
299 X=Y?
300 GTO 00
301 RCL 04
302 X≠0?
303 GTO 01
```

Calculate I

I=0

Program Listings

```
304*LBL 00
305 RCL 06
306 RCL 07
307 *
308 X=0?
309 GTO 04
310 RCL 07
311 RCL 06
312 /
313 CHS
314 RCL 01
315 1/X
316 Y↑X
317 1
318 -
319 GTO 02
320*LBL 01
321 RCL 07
322 RCL 06
323 *
324 X>0?
325 GTO 04
326 RCL 01
327 1/X
328 1
329 +
330 STO 02
331 RCL 06
332 RCL 04
333 *
334 X<0?
335 XEQ 05
336*LBL 12
337 RCL 02
338 LN
339 RCL 01
340 *
341 E↑X-1
342 RCL 02
343 1
344 -
345 X=0?
346 GTO 00
347 /
348 GTO 01
349*LBL 00
350 RCL 01
351*LBL 01
352 STO 08
353 1
354 -
```

Calculate I by simple formula

Begin loop

```
355 RCL 04
356 *
357 RCL 02
358 RCL 01
359 Y↑X
360 RCL 06
361 *
362 +
363 STO 09
364 RCL 01
365 RCL 08
366 -
367 RCL 02
368 1
369 -
370 X=0?
371 GTO 00
372 /
373 RCL 02
374 *
375 RCL 04
376 *
377 GTO 01
378*LBL 00
379 RCL 01
380 1
381 RCL 01
382 -
383 *
384 2
385 /
386*LBL 01
387 RCL 09
388 RCL 01
389 *
390 +
391 RCL 09
392 X<>Y
393 /
394 RCL 07
395 CHS
396 RCL 09
397 /
398 X<>Y
399 Y↑X
400 RCL 02
401 *
402 LASTX
403 X<>Y
404 STO 02
405 X<>Y
```

Program Listings

```
406 %CH
407 ABS
408 1 E-6
409 X<=Y?
410 GTO 12
411 FS? 05
412 XEQ 05
413 CF 05
414 RCL 02
415 1
416 -
417 STO 02
418*LBL 02
419 1 E2
420 *
421 STO 02
422 RTN
423*LBL 05
424 SF 05
425 RCL 02
426 1/X
427 STO 02
428 RCL 07
429 X<> 06
430 STO 07
431 .END.
```

If I not small, repeat loop

Multiply by 100 and store in R_{02}

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	0 or 1	50	
	N		
	I		
	PV		
	PMT		
05	FV	55	
	USED		
	USED		
	USED		
	USED		
0		60	
5		65	
20		70	
25		75	
0		80	
5		85	
0		90	
5		95	

STATUS

SIZE 010 TOT. REG. 99 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		BEGIN	
05		Scratch in I	
21		Printer enable	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

COMPOUND INTEREST SOLUTIONS

PROGRAM REGISTERS NEEDED: 90

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 1 (1 - 3)

ROW 2 (4 - 13)

ROW 3 (14 - 21)

ROW 4 (22 - 28)

ROW 5 (29 - 36)

ROW 6 (36 - 41)

ROW 7 (41 - 47)

ROW 8 (47 - 52)

ROW 9 (52 - 58)

ROW 10 (58 - 62)

ROW 11 (63 - 70)

ROW 12 (70 - 76)

ROW 13 (76 - 82)

ROW 14 (83 - 89)

ROW 15 (89 - 94)

ROW 16 (94 - 99)

ROW 17 (100 - 109)

ROW 18 (110 - 118)

COMPOUND INTEREST SOLUTIONS

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING

ROW 19 (119 - 124)

ROW 20 (125 - 135)

ROW 21 (136 - 148)

ROW 22 (149 - 160)

ROW 23 (161 - 173)

ROW 24 (174 - 179)

ROW 25 (179 - 187)

ROW 26 (188 - 199)

ROW 27 (200 - 205)

ROW 28 (206 - 215)

ROW 29 (216 - 225)

ROW 30 (225 - 233)

ROW 31 (234 - 245)

ROW 32 (246 - 256)

ROW 33 (257 - 267)

ROW 34 (268 - 275)

ROW 35 (276 - 286)

ROW 36 (287 - 299)

COMPOUND INTEREST SOLUTIONS

HEWLETT PACKARD
SOLUTION BOOK:
LENDING SAVING & LEASING



NOTES

NOTES

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

LENDING, SAVING AND LEASING

CONSTANT PAYMENT TO PRINCIPAL LOAN
RULE OF 78'S
AMORTIZATION SCHEDULE
ADD-ON TO APR WITH ODD DAYS
SAVINGS PLAN
INTEREST CONVERSIONS
LEASE WITH ADDITIONAL PAYMENTS IN ADVANCE
SKIPPED PAYMENTS
COMPOUNDING PERIODS DIFFERENT FROM PAYMENT PERIODS
COMPOUND INTEREST SOLUTIONS

00041-90086 Rev. C 11/80 Printed in U.S.A.